TP5

Suites numériques

1 Calcul des termes d'une suite numérique

S'il n'est pas difficile de travailler avec les suites sous Maple, il conviendra avant tout de comprendre comment sont donnés les termes de la suite.

1.1 Suite définie explicitement

Si la suite est définie de façon explicite, du type $u_n = f(n)$, on définira la suite u comme une *expression* dépendant de la variable n.
Ainsi, le calcul des termes consiste simplement à évaluer une expression:

```
> u:=exp(-n)/n^2; eval(u,n=10);
```

$$\frac{1}{100}e^{-10}$$

1.2 Suite définie par une relation de récurrence

Si la suite est définie par récurrence sur plusieurs termes, du type $u_n = f(u_{n-1}, u_{n-2}...)$, on pourra essayer de se ramener à une forme explicite grâce à la commande `rsolve`:

```
> eq:=u(n+1)=2*u(n):  rsolve(eq,u(n));
```

$$u(0)2^n$$

Si on connaît les valeurs initiales, il suffit de remplacer le premier argument de la commande `rsolve` par l'ensemble suivant: $\{equation, conditions\ initiales\}$

```
> eq:=u(n+1)=2*u(n):  rsolve({eq,u(0)=sqrt(2)},u(n));
```

$$\sqrt{2}\,2^n$$

Malheureusement, Maple ne parvient pas toujours à exprimer u_n en fonction de n. En effet, considérons la suite u définie par:

$$u_0 = u_1 = 1 \text{ et } \forall n \geq 1,\ u_{n+1} = u_n^2 + u_{n-1}$$

les mêmes instructions sont inefficaces ici:

```
> eq:=u(n+1)=u(n)^2+u(n-1):  rsolve({eq,u(0)=1,u(1)=1},u(n));
```

$$rsolve(\{u(0) = 1, u(1) = 1, u(n+1) = u(n)^2 + u(n-1)\}, u(n))$$

Dans ce cas, pour calculer les termes de la suite, on devra procéder par <u>itération</u> ou <u>récursivité</u>.

Calcul des termes par la méthode itérative:

On construit une procédure qui prend pour argument une valeur n et renvoie le n-ème terme de la suite, à l'aide d'une boucle `for` et des variables locales:

```
> u:=proc(n)
  local u2,u1,u0,i;
  if n=0 or n=1 then
  u2:=1;
  else
  u1:=1;
  u0:=1;
  for i from 2 to n do
  u2:=u1^2+u0;
  u0:=u1;
  u1:=u2;
  end do;
  end if;
  return u2;
  end proc
```

On calcule le terme souhaité en appelant simplement la procédure u:

```
> u(0);u(1);u(2);u(7);
```

$$1$$
$$1$$
$$2$$
$$290287121823$$

Calcul des termes par la méthode récursive:

La méthode récursive consiste à définir une procédure qui fera appel à elle-même. Plus précisément, pour calculer le n-ème terme de la suite définie plus haut, il est nécessaire d'aller chercher les deux termes précédents u_{n-1} et u_{n-2}, eux-mêmes nécessitant le calcul de termes précédents... Maple lance donc les procédures au fur et à mesure jusqu'à obtenir u_1 et u_0, puis il remonte tous ses calculs pour donner la valeur de u_n:

```
> u:=proc(n)
  local resultat;
  if n=0 or n=1 then
  resultat:=1;
  else resultat:=u(n-1)^2+u(n-2);
  end if;
  return resultat;
  end proc
```

Cependant il s'agira de faire bien attention à ajouter l'option `remember` dans notre procédure. En effet, lorsque Maple remonte la pile de ses calculs pour trouver u_n, il calcule toutes les valeurs u_k ($k < n$) dont il a besoin et cela même s'il les a déjà calculées. Cette option nous permet par conséquent de forcer Maple à remonter ses calculs en se rappelant des valeurs rencontrées:

```
> u:=proc(n)
  option remember;
  local resultat;
  if n=0 or n=1 then
  resultat:=1;
  else resultat:=u(n-1)^2+u(n-2);
  end if;
  return resultat;
  end proc
```

De la même façon, on calcule le terme souhaité en appelant la procédure u:

```
> u(0);u(1);u(2);u(7);
```

$$1$$
$$1$$
$$2$$
$$290287121823$$

Exercice 1 On définit la suite (u_n) par:

$$u_0 = 1, u_1 = 1 \text{ et } \forall n \geq 2,\ u_n = u_{n-1} + u_{n-2}$$

Déterminer u_n en fonction de n. En déduire sa limite quand $n \to \infty$.

Exercice 2 On définit la suite (v_n) par:

$$v_0 = 0.5 \text{ et } \forall n \geq 1,\ v_n = \frac{2}{3 - v_{n-1}}$$

1. Ecrire une procédure **récursive** qui, pour un entier n donné, calcule v_n.
2. Calculer différentes valeurs v_n, puis établir une conjecture sur sa convergence.

Exercice 3 On considère la suite (w_n) définie par: $w_0 = 1$ et $\forall n \geq 1,\ w_n = w_{n-1}^2 + 1$.

1. Ecrire une procédure **itérative** qui, pour un entier n donné, calcule w_n. Etablir une conjecture sur sa convergence.
2. Modifier votre programme afin qu'il ne renvoie pas seulement w_n, mais une liste avec toutes les valeurs $w_0, \dots, w_n$.
3. *Les prochaines questions sont à résoudre indépendamment de Maple.*
   (a) Montrer que pour tout $n \in \mathbb{N}$, $w_n \geq 1$.
   (b) En déduire que la suite (w_n) est croissante, et que $lim_{n\to+\infty} w_n = +\infty$.

2 Calcul et approximation de la limite d'une suite

Si la suite est définie explicitement, la limite pourra être déterminée par la commande `limit`:

```
> limit(n^n/n!,n=infinity)
```

$$\infty$$

Dans le cas contraire, on s'intéressera à sa limite de deux façons:

- soit par approximation de celle-ci en calculant les termes de la suite pour n assez grand,
- soit, quand elle est connue, en étudiant le module de la différence: $|u_n - l|$.

Dans ces deux cas, on observera la vitesse de convergence de la suite, c'est à dire la rapidité avec laquelle la suite tend vers sa limite.

En guise d'exemple, considérons la suite réelle définie par: $u_1 = 1$ et $\forall n \geq 2,\ u_n = u_{n-1} + \dfrac{1}{n^2}$.

2.1 Approximation de la limite

On commence par construire la procédure pour calculer les termes de la suite:

```
> u:=proc(n)
  option remember;
  local resultat;
  if n=1 then
  resultat:=1;
  else resultat:=u(n-1)+1/n^2;
  end if;
  return resultat;
  end proc
```

On peut alors afficher la séquence des valeurs de la suite pour n assez grand:

```
> seq(evalf(u(1000+i)),i=1..1000)
```

1.643935565,1.643936561,1.643937555,1.643938547...

Ainsi, on remarque que cette suite converge lentement ($n \sim 2000$) vers un nombre dont l'écriture décimale est:

1.644...

2.2 Vitesse de convergence

La suite (u_n) désigne en fait une somme bien connue:

$$1 + \frac{1}{2^2} + \frac{1}{3^2} + \frac{1}{4^2} + \dots + \frac{1}{n^2} = \sum_{k=1}^{n} \frac{1}{k^2} \qquad \text{dont la limite est } \frac{\pi^2}{6}.$$

Nous allons construire une procédure qui détermine la première valeur n pour laquelle $|u_n - \dfrac{\pi^2}{6}| < 10^{-a}$: celle-ci nous donnera donc le rang à partir duquel u_n approche sa limite avec une précision de l'ordre de 10^{-a}.

```
> approx:=proc(a)
  local n,u;
  n:=1; u:=1;
  while evalf(abs(u-Pi^2/6))>=10^(-a) do
  n:=n+1; u:=u+1/n^2;
  end do;
  return n;
  end proc
```

Par conséquent, si on appelle cette procédure pour $a = 4$, on peut alors constater la lenteur de la convergence; en effet, la suite approche sa limite à 10^{-4} près à partir du rang $n = 10000$:

```
> approx(4)
```

$$10000$$

Exercice 4 On considère la suite (u_n) définie par $u_0 = 2$ et $\forall n \geq 1,\ u_n = f(u_{n-1})$ avec $f : x \mapsto \frac{1}{2}(x + \frac{2}{x})$.

1. *Les prochaines questions sont à résoudre indépendamment de Maple.*
   (a) Montrer que pour tout $n \in \mathbb{N}$, u_n est bien défini et $u_n \geq \sqrt{2}$.
   (b) Montrer que la suite (u_n) est convergente et déterminer sa limite.
2. Ecrire une procédure `approx(a)` qui renvoie le rang à partir duquel la suite approche sa limite à 10^{-a} près.

Exercice 5 Soit N un entier. On définit la suite de Syracuse associée à l'entier N par:

$$u_1 = N \text{ et } \forall n \geq 2,\ u_n = \begin{cases} \frac{u_{n-1}}{2}\ (si\ \ u_{n-1}\ \ pair) \\ \frac{3u_{n-1}+1}{2}\ (si\ \ u_{n-1}\ \ impair) \end{cases}$$

1. Ecrire une procédure **itérative** `syr1(N,n)` qui renvoie une liste dans laquelle seront affichées les valeurs $u_1, \ldots, u_n$ avec $u_1 = N$.
2. Calculer `syr1(15,50)` et `syr1(127,50)`.

On appelle temps de vol le plus petit indice p tel que $u_p = 1$.

3. Ecrire une procédure itérative `syr2(N)` qui calcule les valeurs u_n tant que $u_n \neq 1$, puis en sortie renvoie le temps de vol obtenu.
4. Déterminer le temps de vol pour $N = 15$ et $N = 222$.